
2 Corrections required for 8.1.2.11 and f_EUTRA_IdleUpdated_Step14_15
2.1 Corrections to function ‘f_TC_8_1_2_11_EUTRA ()’

	Function name
	f_TC_8_1_2_11_EUTRA

	Reason for change
	1.  UM DRB Data must be discard in order to avoid unfair verdicts.

	Summary of change
	1.  UM  Configurationd set to Discard UL Data 

	TTCN module
	RRC_EmergencyCall.ttcn.

	MCC160 Comment
	


Before change

function f_TC_8_1_2_11_EUTRA() runs on EUTRA_PTC

  { /* RRC Connention establishment of emergency call */

    f_EUTRA_Init(c1);

    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_Cell1, 1, 1); // @sic R5-131470 sic@

    //Preamble to enter UE in E-UTRA RRC_IDLE (State 2)

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE, -, 2);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_2_11_Body();

    f_EUTRA_TestBody_Set(false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }

After change

function f_TC_8_1_2_11_EUTRA() runs on EUTRA_PTC

  { /* RRC Connention establishment of emergency call */

    f_EUTRA_Init(c1);

    // Create cell and send out system information

    f_EUTRA_CellConfig_DRB_Common(eutra_Cell1, 1, 1,DISCARD_UM_DATA); // @sic R5-131470 sic@

    //Preamble to enter UE in E-UTRA RRC_IDLE (State 2)

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE, -, 2);

    f_EUTRA_TestBody_Set(true);

    fl_TC_8_1_2_11_Body();

    f_EUTRA_TestBody_Set(false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }  

2.2 Corrections to function ‘f_EUTRA_IdleUpdated_Step14_15()’.

	Function name
	f_EUTRA_IdleUpdated_Step14_15

	Reason for change
	Network feature indication supporting emergency call is mandatory for Emergency call test cases which are rel9 feature according to TS 24.301 clause 5.5.1.2.4.

	Summary of change
	Set to supported IMS and LBS rel9 feature.

	TTCN module
	EUTRA_CommonProcedures.ttcn

	MCC160 Comment
	


Before change

function f_EUTRA_IdleUpdated_Step14_15(EUTRA_CellId_Type p_CellId,

                                         EUTRA_ATTACH_Type p_ForcedAttach,

                                         template (omit) AdditionalUpdateType p_AdditionalUpdateType,

                                         ProcedureTransactionIdentifier p_EPS_TI,

                                         boolean p_IPv4AllocationViaNasFlag,

                                         template (omit) ProtocolConfigOptions p_PcoToUE,

                                         template (omit) AccessPointName p_APN,

                                         PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE,

                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                         template (omit) GPRS_Timer p_T3402 := omit,

                                         template (omit) GPRS_Timer p_T3423 := omit,

                                         template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                         integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@

                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice) runs on EUTRA_PTC

  { /* @sic R5-112600: IP address allocation sic@ */

    /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@ */

    /* @sic R5-120234, R5-120541, R5-120542 sic@ */

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit) EPS_NetworkFeatureSupport v_NtwkFeatSupport := p_NtwkFeatSupport;

    var template (omit) EmergNumList v_EmergNumList := omit;

    var EmergencyNumList v_EmergencyNumList;

    var template (omit) MobileIdentity v_MSId := f_GetMSId (p_ForcedAttach); // @sic R5-103673 sic@

    var template (value) PDN_Address v_PDN_Address := f_EUTRA_GetPDNAddress (p_IPv4AllocationViaNasFlag, -, p_PDN_TypeToBeUsed);

    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */

      select (p_PDN_TypeToBeUsed) {

        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed

        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed

        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }

      }

    }

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release (testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

       v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

      }

      case else {}

    }

    if (p_NoOfEmergencyNumbers > 0) {

      v_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice_EMCBS_EPCLCS;

      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_UT_GetEmergencyNumbers(UT));

      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers

      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);

    }

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,

                                                                  tsc_EpsDefaultBearerId,

                                                                  tsc_RRC_TI_Def,

                                                                  p_EPS_TI,

                                                                  v_AttachType,

                                                                  v_PDN_Address,

                                                                  v_Guti,

                                                                  cs_TAIListNonConsecutive_lv (v_PLMN, {bit2oct(v_TAC)}),

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  v_ESM_Cause, // @sic R5-104703 sic@

                                                                  p_APN,

                                                                  p_PcoToUE,

                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                  p_T3412,

                                                                  p_T3402,

                                                                  p_T3423,

                                                                  p_EquivalentPlmnList,

                                                                  v_EmergNumList,        // @sic R5s130681 sic@

                                                                  v_NtwkFeatSupport);    // @sic R5s130681 sic@

  }

    }

After change

function f_EUTRA_IdleUpdated_Step14_15(EUTRA_CellId_Type p_CellId,

                                         EUTRA_ATTACH_Type p_ForcedAttach,

                                         template (omit) AdditionalUpdateType p_AdditionalUpdateType,

                                         ProcedureTransactionIdentifier p_EPS_TI,

                                         boolean p_IPv4AllocationViaNasFlag,

                                         template (omit) ProtocolConfigOptions p_PcoToUE,

                                         template (omit) AccessPointName p_APN,

                                         PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE,

                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                         template (omit) GPRS_Timer p_T3402 := omit,

                                         template (omit) GPRS_Timer p_T3423 := omit,

                                         template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                         integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@

                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice) runs on EUTRA_PTC

  { /* @sic R5-112600: IP address allocation sic@ */

    /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@ */

    /* @sic R5-120234, R5-120541, R5-120542 sic@ */

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit) EPS_NetworkFeatureSupport v_NtwkFeatSupport := p_NtwkFeatSupport;

    var template (omit) EmergNumList v_EmergNumList := omit;

    var EmergencyNumList v_EmergencyNumList;

    var template (omit) MobileIdentity v_MSId := f_GetMSId (p_ForcedAttach); // @sic R5-103673 sic@

    var template (value) PDN_Address v_PDN_Address := f_EUTRA_GetPDNAddress (p_IPv4AllocationViaNasFlag, -, p_PDN_TypeToBeUsed);

    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */

      select (p_PDN_TypeToBeUsed) {

        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed

        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed

        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }

      }

    }

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release (testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

       v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

       v_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice_EMCBS_EPCLCS; 

      }

      case else {}

    }

    if (p_NoOfEmergencyNumbers > 0) {

      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_UT_GetEmergencyNumbers(UT));

      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers

      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);

    }

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,

                                                                  tsc_EpsDefaultBearerId,

                                                                  tsc_RRC_TI_Def,

                                                                  p_EPS_TI,

                                                                  v_AttachType,

                                                                  v_PDN_Address,

                                                                  v_Guti,

                                                                  cs_TAIListNonConsecutive_lv (v_PLMN, {bit2oct(v_TAC)}),

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  v_ESM_Cause, // @sic R5-104703 sic@

                                                                  p_APN,

                                                                  p_PcoToUE,

                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                  p_T3412,

                                                                  p_T3402,

                                                                  p_T3423,

                                                                  p_EquivalentPlmnList,

                                                                  v_EmergNumList,        // @sic R5s130681 sic@

                                                                  v_NtwkFeatSupport);    // @sic R5s130681 sic@

  }

    }  

